AN ABSTRACT OF THE THESIS OF

John Allen Dowling for the Master of Arts in Mathematics presented on July 27, 1983

Title: Graphing Conic Sections on the Microcomputer

Abstract approved: John W. Carlson

This thesis investigates graphing conic sections on the microcomputer. The different entry points of the programs allows wide use as a teaching aid to examine many different properties of the conic sections. The microcomputer acts as a time saving aid to examine these properties.

GRAPHING CONIC SECTION ON THE MICROCOMPUTER

A Thesis

Presented to

the Department of Mathematics

EMPORIA STATE UNIVERSITY

In Partial Fulfillment

of the Requirements for the Degree

Master of Arts

By

John A. Dowling

July, 1983

John W. Cuskey

Approved for the Major Department

Harold E. Sweet

Approved for the Graduate Council

ACKNOWLEDGMENTS

I would like to extend my deep appreciation to Dr. John Carlson for his assistance and encouragement and to all my professors at Emporia State University who made this study possible.

Acknowledgment and appreciation are also expressed to my wife who was understanding and thoughtful while this study was in progress.

TABLE OF CONTENTS

TABLE OF PROGRAMS

1. DEFINITIONS AND TERMS

The use of microcomputers in the classroom is an area that is growing quite rapidly. This thesis will consider the application of the microcomputer to graph conic sections from different entry points. These will include: conics in standard position, conics with a translation of axis, conics with a rotation of axis, a general program to graph conics, and a program to investigate the eccentricity of conics.

The following definitions will be used to establish equations in the latter chapters.

DEFINITION 1.1 The set of points P equidistant from a fixed point F and a line l is called a parabola. See FIG. 1.1.

FIG. 1.1



The line l is called the directrix of the parabola and the point F is called the focus. The line through F which is perpendicular to l is called the axis of the parabola. The point at which the axis intersects the parabola is called the vertex.

DEFINITION 1.2 The set of all points P such that

$$d(P,F_1) + d(P,F_2) = k$$

is called an ellipse, where fixed points F_1 and F_2 and called foci. See FIG. 1.2.

FIG. 1.2



DEFINITION 1.3 The set of points P such that

$$|d(P,F_1)-d(P,F_2)| = k$$

is called a hyperbola, where fixed points F_1 and F_2 are called the foci. See FIG. 1.3.

FIG. 1.3



2. PARABOLAS

To graph any conic on the microcomputer you must determine certain parameters that define the conic and then convert the information to BASIC. In the case of the parabola, by using the definition and making the restriction that the vertex is at the origin the resulting equation can be developed.

$$d(P,F)=d(P,l)$$

$$d(P,F)=\sqrt{x^2+(y-c)^2}, \text{ and } d(P,l)=|y+c|$$

$$\sqrt{x^2+(y-c)^2} = |y+c|$$

$$x^2+(y-c)^2=(y+c)^2$$

$$x^2+y^2-2cy+c^2=y^2+2cy+c^2$$

EQ. 2.1.1 $$x^2=4cy$$ (See FIG. 2.1.1.)

FIG. 2.1.1



From FIG. 2.1.1 we see that for the equation $x^2=4cy$ the graph is a parabola with vertex at the origin and focus (0,c). If the vertex remains at the origin and the focus is placed at (c,0) the equation takes the form of

EQ. 2.1.2 $y^2=4cx$ (See FIG. 2.1.2.)

FIG. 2.1.2



In BASIC EQ. 2.1.1 becomes:

```
Y = X^2/(4*C)
```

and EQ. 2.1.2 becomes:

```
Y1 = 2*SQR(C*X)
Y2 = -Y1
```

If a translation of axis is needed, then new equations are used to represent the parabola:

EQ. 2.2.1 $(x-x0)^2 = 4c(y-y0)$ See FIG. 2.2.1

EQ. 2.2.2 $(y-y0)^2 = 4c(x-x0)$ See FIG. 2.2.2

FIG. 2.2.1



FIG. 2.2.2



In BASIC these become:

```
Y = (X-XO)^2/(4*C)+YO          for EQ. 2.2.1,
```

and

```
Y1 = 2*SQR(C*(X-XO))+YO
Y2 =-2*SQR(C*(X-XO))+YO        for EQ. 2.2.2.
```

Finally, to rotate a parabola a given angle θ consider FIG. 2.3.1.

FIG. 2.3.1



From the figure $\cos\theta = x/r$ and $\sin\theta = y/r$. Thus $x=r\cos\theta$ and $y=r\sin\theta$.

Consider now a rectangular coordinate system Oxy. If we rotate this system θ degrees about the origin, we obtain a new coordinate system OXY [3. pp. 432-433]. See FIG. 2.3.2.

FIG. 2.3.2



A point will now have two pair of rectangular coordinates:
(x,y) in the Oxy system and (X,Y) in the OXY system. Here we investigate the relationship between (x,y) and (X,Y) with P in FIG. 2.3.3.

FIG. 2.3.3



$x = r\cos(\theta+\beta)$, $\quad y = r\sin(\theta+\beta)$

and

$X = r\cos(\beta)$, $\quad Y = r\sin(\beta)$.

Since $\cos(\theta+\beta) = \cos\theta\cos\beta - \sin\theta\sin\beta$,

$\sin(\theta+\beta) = \sin\theta\cos\beta + \cos\theta\sin\beta$,

we have $x = r\cos(\theta+\beta) = (\cos\theta)r\cos\beta - (\sin\theta)r\sin\beta$,

$y = r\sin(\theta+\beta) = (\sin\theta)r\cos\beta + (\cos\theta)r\sin\beta$,

and therefore

EQ. 2.3.1 $\quad x = \cos\theta\, X - \sin\theta\, Y$

and

EQ. 2.3.2 $\quad y = \sin\theta\, X + \cos\theta\, Y$.

Using EQ. 2.2.1 and EQ. 2.3.1 we can now consider graphing parabolas with a rotation of θ degrees. Since (X,Y) are the ordered pairs before a rotation, the new ordered pairs in BASIC would be:

```
XP = COS(TH)*X-SIN(TH)*Y

YP = SIN(TH)*X-COS(TH)*Y
```

The following is a complete listing of PROG. 2.1 to graph parabolas given the following inputs: C,XO,YO, and, TH.

PROG. 2.1

```
100  REM  PROGRAM TO GRAPH PARABOLAS
110  TEXT
120  HOME
130  REM
140  REM  ACCEPT TYPE OF PARABOLA
150  REM
160  PRINT "INPUT TYPE OF PARABOLA"
170  PRINT
180  PRINT "    <1>  (X-XO)^2 = 4C(Y-YO)
190  PRINT
200  PRINT "    <2>  (Y-YO)^2 = 4C(X-XO)
210  PRINT
220  INPUT "ENTER THE NUMBER OF YOUR CHOICE ";CC
230  PRINT
240  INPUT "INPUT THE VALUE OF C>>>";C
250  INPUT "INPUT THE VALUE OF XO>>";XO
260  INPUT "INPUT THE VALUE OF YO>>";YO
270  INPUT "INPUT THE ROTATION IN DEGREES ";TH
280 TT = TH
290 TH = TH / 57.2958: REM  CONVERT TO RADIANS
300  ON CC GOTO 330,360
310  PRINT "BAD RESPONSE": FOR J = 1 TO 1000: NEXT J
320  GOTO 120
330  DEF  FN Y1(X) =  SIN (TH) * X +  COS (TH) * (X ^ 2 / (4
 * C))
340  DEF  FN X1(X) =  COS (TH) * X -  SIN (TH) * (X ^ 2 / (4
 * C))
350  GOTO 390
360  DEF  FN Y1(X) =  SIN (TH) * X +  COS (TH) * (2 *  SQR (
C * X))
370  DEF  FN Y2(X) =  SIN (TH) * X +  COS (TH) * ( - 2 *  SQ
R (C * X))
380  DEF  FN X1(X) =  COS (TH) * X -  SIN (TH) * (2 *  SQR (
C * X))
```

PROG. 2.1 CONT.

```
390  DEF  FN X2(X) =  COS (TH) * X -  SIN (TH) * ( - 2 *  SQ
R (C * X))
400  REM
410  REM  GRAPHING ROUTINE
420  REM
430  REM  ASSIGN SCREEN LIMITS
440 HL = 279
450 VL = 159
460 VS = 80
470 HS = 140
480  REM
490  REM ACCEPT VALUES FOR XMIN, XMAX, YMIN, YMAX
500  REM
510  INPUT "USE DEFAULT VALUES FOR X AND Y RANGES?";ANS$
520  IF  LEFT$ (ANS$,1) = "N" THEN 550
530 X1 =  - 100:X2 = 100:Y1 =  - 100:Y2 = 100
540  GOTO 590
550  INPUT "XMIN = ";X1
560  INPUT "XMAX = ";X2
570  INPUT "YMIN = ";Y1
580  INPUT "YMAX = ";Y2
590  REM
600  REM   ASSIGN RANGE MULTIPLIERS AND DELTA X
610  REM
620 XM = 1
630 YM =  - .85
640 DX = 1
650  REM
660  REM  DRAW COORDINATE AXIS
670  REM
680  HGR
690  HCOLOR= 6
700  HPLOT 140,0 TO 140,159
710  HPLOT 0,80 TO 279,80
720  REM
730  REM  PLOTTING LOOP
740  REM
750  HCOLOR= 3
760  FOR X = X1 TO X2 STEP DX
770  IF CC = 2 AND C * X < 0 THEN 890
```

PROG. 2.1 CONT.

```
780 XP = XM *  FN X1(X) + HS + XO
790 YP = YM *  FN Y1(X) + VS - YO
800  IF XP > HL OR XP < 0 THEN 830
810  IF YP > VL OR YP < 0 THEN 830
820  HPLOT XP,YP
830  IF CC = 1 THEN 890
840 YN = YM *  FN Y2(X) + VS - YO
850 XN = XM *  FN X2(X) + HS + XO
860  IF YN > VL OR YN < 0 THEN 890
870  IF XN > HL OR XN < 0 THEN 890
880  HPLOT XN,YN
890  NEXT X
900  END
```

3. ELLIPSES

The next conic to be considered is the ellipse. Using DEFINITION 1.2 and setting k = 2a with the restriction the foci are placed along the x-axis at equal distances from the origin (See FIG. 3.1.) the resulting equation can be developed.

$d(P,F_1) + d(p,F_2) = 2a$

$\sqrt{(x+c)^2 + y^2} + \sqrt{(x-c)^2 + y^2} = 2a$

$\sqrt{(x+c)^2 + y^2} = 2a - \sqrt{(x-c)^2 + y^2}$

$(x+c)^2+y^2=4a^2-4a\sqrt{(x-c)^2+y^2}+(x-c)^2+y^2$

$4xc = 4a^2 - 4a\sqrt{(x-c)^2 + y^2}$

$-xc/a + a^2 = \sqrt{(x-c)^2 + y^2}$

$a^2-2xc+(x^2c^2)/a^2=x^2 - 2xc + c^2 + y^2$

$a^2 - c^2 = x^2 (a^2 - c^2)/a^2 + y^2$

$1 = x^2/a^2 + y^2/(a^2 - c^2)$

let $b = \sqrt{a^2 - c^2}$

EQ. 3.1 $\quad 1 = x^2/a^2 + y^2/b^2$

FIG. 3.1



In a similar manner EQ. 3.2 can be derived where the foci are on the y axis.

EQ. 3.2 $1 = x^2/b^2 + y^2/a^2$

In BASIC EQ. 3.1 becomes

```
Y1 = B*SQR(1-X^2/(A^2))
Y2 = -Y1
```

and EQ. 3.2 becomes

```
Y1 = A*SQR*(1-X^2/(B^2))
Y2 = -Y1
```

If a translation of axis is desired, then using EQ. 3.1 the new equation would be

EQ. 3.2.1 $$\frac{(X-XO)^2}{A^2} + \frac{(Y-YO)^2}{B^2} = 1$$

and using EQ. 3.2

EQ. 3.2.2 $$\frac{(X-XO)^2}{B^2} + \frac{(Y-YO)^2}{A^2} = 1$$

In BASIC EQ. 3.2.1 takes the form of

```
Y1 = B*SQR(1-(X-XO)^2/(A^2))+YO
Y2 =-B*SQR(1-(X-XO)^2/(A^2))+YO
```

and EQ. 3.2.2 would be

```
Y1 = A*SQR(1-(X-XO)^2/(B^2))+YO
Y2 =-A*SQR(1-(X-XO)^2/(B^2))+YO
```

Finally to achieve a rotation of θ degrees about a fixed point (XO,YO) we would simply reuse EQ. 2.3.1 and EQ. 2.3.2 with the equations for the ellipse.

The following is a complete listing of PROG. 3.1 which will graph ellipses given inputs of A,B,XO,YO, and, TH.

PROG. 3.1

```
100  REM  PROGRAM TO GRAPH ELLIPSES
110  TEXT
120  HOME
130  REM
140  REM  ACCEPT TYPE OF ELLIPSE
150  REM
160  PRINT "INPUT TYPE OF ELLIPSE"
170  PRINT
180  PRINT "          (X-XO)^2    (Y-YO)^2"
190  PRINT "  <1>    -------- + -------- = 1"
200  PRINT "            A^2         B^2"
210  PRINT
220  PRINT "          (X-XO)^2    (Y-YO)^2"
230  PRINT "  <2>    -------- + -------- = 1"
240  PRINT "            B^2         A^2"
250  PRINT
260  INPUT "ENTER THE NUMBER OF YOUR CHOICE ";CC
270  PRINT
280  INPUT "INPUT THE VALUE OF A>>>";A
290  INPUT "INPUT THE VALUE OF B>>>";B
300  INPUT "INPUT THE VALUE OF XO>>";XO
310  INPUT "INPUT THE VALUE OF YO>>";YO
320  INPUT "INPUT THE ROTATION IN DEGREES ";TH
330 TT = TH
340 TH = TH / 57.2958: REM  CONVERT TO RADIANS
350  ON CC GOTO 380,430
360  PRINT "BAD RESPONSE": FOR J = 1 TO 1000: NEXT J
370  GOTO 120
380  DEF  FN Y1(X) =  SIN (TH) * X +  COS (TH) * (B *  SQR (
1 - (X / A) ^ 2))
390  DEF  FN Y2(X) =  SIN (TH) * X +  COS (TH) * ( - B *  SQ
R (1 - (X / A) ^ 2))
```

PROG. 3.1 CONT.

```
400  DEF  FN X1(X) =  COS (TH) * X -  SIN (TH) * (B *  SQR (
1 - (X / A) ^ 2))
410  DEF  FN X2(X) =  COS (TH) * X -  SIN (TH) * ( - B *  SQ
R (1 - (X / A) ^ 2))
420  GOTO 470
430  DEF  FN Y1(X) =  SIN (TH) * X +  COS (TH) * (A *  SQR (
1 - (X / B) ^ 2))
440  DEF  FN Y2(X) =  SIN (TH) * X +  COS (TH) * ( - A *  SQ
R (1 - (X / B) ^ 2))
450  DEF  FN X1(X) =  COS (TH) * X -  SIN (TH) * (A *  SQR (
1 - (X / B) ^ 2))
460  DEF  FN X2(X) =  COS (TH) * X -  SIN (TH) * ( - A *  SQ
R (1 - (X / B) ^ 2))
470  REM
480  REM  GRAPHING ROUTINE
490  REM
500  REM  ASSIGN SCREEN LIMITS
510 HL = 279
520 VL = 159
530 VS = 80
540 HS = 140
550  REM
560  REM ACCEPT VALUES FOR XMIN, XMAX, YMIN, YMAX
570  REM
580  INPUT "USE DEFAULT VALUES FOR X AND Y RANGES?";ANS$
590  IF  LEFT$ (ANS$,1) = "N" THEN 620
600 X1 =  - 100:X2 = 100:Y1 =  - 100:Y2 = 100
610  GOTO 660
620  INPUT "XMIN = ";X1
630  INPUT "XMAX = ";X2
640  INPUT "YMIN = ";Y1
650  INPUT "YMAX = ";Y2
660  REM
670  REM   ASSIGN RANGE MULTIPLIERS AND DELTA X
680  REM
690 XM = 1
700 YM =  - .85
710 DX = 1
720  REM
730  REM  DRAW COORDINATE AXIS
740  REM
750  HGR
760  HCOLOR= 6
770  HPLOT 140,0 TO 140,159
780  HPLOT 0,80 TO 279,80
790  REM
800  REM  PLOTTING LOOP
```

PROG. 3.1 CONT.

```
810  REM
820  HCOLOR= 3
830  FOR X = X1 TO X2 STEP DX
840  IF CC = 1 AND 1 - (X / A) ^ 2 < 0 THEN 960
850  IF CC = 2 AND 1 - (X / B) ^ 2 < 0 THEN 960
860 XP = XM *  FN X1(X) + HS + XO
870 YP = YM *  FN Y1(X) + VS - YO
880  IF XP > HL OR XP < 0 THEN 960
890  IF YP > VL OR YP < 0 THEN 910
900  HPLOT XP,YP
910 YN = YM *  FN Y2(X) + VS - YO
920 XN = XM *  FN X2(X) + HS + XO
930  IF YN > VL OR YN < 0 THEN 960
940  IF XN > HL OR XN < 0 THEN 960
950  HPLOT XN,YN
960  NEXT X
970  END
```

4. HYPERBOLAS

The final conic to be considered is the hyperbola. Using DEFINITION 1.3 and again setting k=2a with the restriction the foci are placed along the x-axis at equal distance from the origin (See FIG. 4.1.) the resulting equation can be developed:

$$d(P,F_1) - d(P,F_2) = 2a$$

then in a similiar manner as with the ellipse,

EQ. 4.1.1
$$\frac{x^2}{a^2} - \frac{y^2}{b^2} = 1$$

and when the foci are placed along the y-axis at equal distance from the origin (See FIG. 4.2.) the equation is:

EQ. 4.1.2
$$\frac{x^2}{b^2} - \frac{y^2}{a^2} = -1$$

FIG. 4.1



FIG. 4.2



The next topic to consider is a translation of axis for the hyperbola. Using EQ. 4.1.1 with center (xO,yO) the resulting equation becomes:

$$\text{EQ. 4.2.1} \qquad \frac{(x-xO)^2}{a^2} - \frac{(y-yO)^2}{b^2} = 1$$

and EQ. 4.1.2 would be

$$\text{EQ. 4.2.2} \qquad \frac{(x-xO)^2}{b^2} - \frac{(y-yO)^2}{a^2} = -1$$

In BASIC EQ. 4.2.1 becomes

```
Y1 = B*SQR(1+(X-XO)^2/(A^2))+YO

Y2 =-B*SQR(1+(X-XO)^2/(A^2))+YO
```

and EQ. 4.2.2 would be

```
Y1 = A*SQR(1+(X-XO)^2/(B^2))+YO

Y2 =-A*SQR(1+(X-XO)^2/(B^2))+YO
```

Finally to achieve a rotation of θ degrees about a fixed point (XO,YO) we would in a similar manner as with the ellipse reuse EQ. 2.3.1 and EQ. 2.3.2 with the equations for the hyperbola.

The following is a complete listing of PROG. 4.1 which will graph hyperbolas given inputs of A,B,XO,YO, and, TH.

PROG. 4.1

```
100  REM  PROGRAM TO GRAPH HYPERBOLAS
110  TEXT
120  HOME
130  REM
140  REM  ACCEPT TYPE OF HYPERBOLA
150  REM
160  PRINT "INPUT TYPE OF HYPERBOLA"
170  PRINT
180  PRINT "          (X-XO)^2   (Y-YO)^2"
190  PRINT "  <1>    -------- - -------- = 1"
200  PRINT "            A^2        B^2"
210  PRINT
220  PRINT "          (X-XO)^2   (Y-YO)^2"
230  PRINT "  <2>    -------- - -------- = 1"
240  PRINT "            B^2        A^2"
250  PRINT
260  INPUT "ENTER THE NUMBER OF YOUR CHOICE ";CC
270  PRINT
280  INPUT "INPUT THE VALUE OF A>>>";A
290  INPUT "INPUT THE VALUE OF B>>>";B
300  INPUT "INPUT THE VALUE OF XO>>";XO
310  INPUT "INPUT THE VALUE OF YO>>";YO
320  INPUT "INPUT THE ROTATION IN DEGREES ";TH
```

PROG. 4.1 CONT.

```
330 TT = TH
340 TH = TH / 57.2958: REM  CONVERT TO RADIANS
350  ON CC GOTO 380,430
360  PRINT "BAD RESPONSE": FOR J = 1 TO 1000: NEXT J
370  GOTO 120
380  DEF  FN Y1(X) =  SIN (TH) * X +  COS (TH) * (B *  SQR (
(X / A) ^ 2 - 1))
390  DEF  FN Y2(X) =  SIN (TH) * X +  COS (TH) * ( - B *  SQ
R ((X / A) ^ 2 - 1))
400  DEF  FN X1(X) =  COS (TH) * X -  SIN (TH) * (B *  SQR (
(X / A) ^ 2 - 1))
410  DEF  FN X2(X) =  COS (TH) * X -  SIN (TH) * ( - B *  SQ
R ((X / A) ^ 2 - 1))
420  GOTO 390
430  DEF  FN Y1(X) =  SIN (TH) * X +  COS (TH) * (A *  SQR (
(X / B) ^ 2 - 1))
440  DEF  FN Y2(X) =  SIN (TH) * X +  COS (TH) * ( - A *  SQ
R ((X / B) ^ 2 - 1))
450  DEF  FN X1(X) =  COS (TH) * X -  SIN (TH) * (A *  SQR (
(X / B) ^ 2 - 1))
460  DEF  FN X2(X) =  COS (TH) * X -  SIN (TH) * ( - A *  SQ
R ((X / B) ^ 2 - 1))
470  REM
480  REM  GRAPHING ROUTINE
490  REM
500  REM  ASSIGN SCREEN LIMITS
510 HL = 279
520 VL = 159
530 VS = 80
540 HS = 140
550  REM
560  REM ACCEPT VALUES FOR XMIN, XMAX, YMIN, YMAX
570  REM
580  INPUT "USE DEFAULT VALUES FOR X AND Y RANGES?";ANS$
590  IF  LEFT$ (ANS$,1) = "N" THEN 620
600 X1 =  - 100:X2 = 100:Y1 =  - 100:Y2 = 100
610  GOTO 660
620  INPUT "XMIN = ";X1
630  INPUT "XMAX = ";X2
640  INPUT "YMIN = ";Y1
650  INPUT "YMAX = ";Y2
660  REM
670  REM   ASSIGN RANGE MULTIPLIERS AND DELTA X
680  REM
690 XM = 1
700 YM =  - .85
710 DX = 1
```

PROG. 4.1 CONT.

```
720  REM
730  REM  DRAW COORDINATE AXIS
740  REM
750  HGR
760  HCOLOR= 6
770  HPLOT 140,0 TO 140,159
780  HPLOT 0,80 TO 279,80
790  REM
800  REM  PLOTTING LOOP
810  REM
820  HCOLOR= 3
830  FOR X = X1 TO X2 STEP DX
840  IF CC = 1 AND (X / A) ^ 2 - 1 < 0 THEN 960
850  IF CC = 2 AND (X / B) ^ 2 - 1 < 0 THEN 960
860 XP = XM *  FN X1(X) + HS + XO
870 YP = YM *  FN Y1(X) + VS - YO
880  IF XP > HL OR XP < 0 THEN 960
890  IF YP > VL OR YP < 0 THEN 910
900  HPLOT XP,YP
910 YN = YM *  FN Y2(X) + VS - YO
920 XN = XM *  FN X2(X) + HS + XO
930  IF YN > VL OR YN < 0 THEN 960
940  IF XN > HL OR XN < 0 THEN 960
950  HPLOT XN,YN
960  NEXT X
970  END
```

GENERAL EQUATION

The equation of any conic section can be expressed as a second degree equation:

Eq. 5.1 $Ax^2+Bxy+Cy^2+Dx+Ey+F=0$ A,B,C not all 0.

Consider the following 3 cases that evolve from the discrimant of EQ. 5.1.

Case 1. If $B^2 - 4AC < 0$ then the graph is an ellipse, a circle, a point, or empty.

Case 2. If $B^2 - 4AC = 0$ then the graph is a parabola, a line, a pair of lines, or empty.

Case 3. If $B^2 - 4AC > 0$ then the graph is a hyperbola or a pair of intersecting lines.

Graphs 5.1.1 to 5.3.4 at the end of the chapter illustrate examples of the above cases.

The following is the algebraic work that was involved to transform EQ. 5.1 to BASIC:

$$Ax^2 + Bxy + Cy^2 + Dx + Ey + F = 0$$

$$Cy^2 + Bxy + Ey = -Ax^2 - Dx - F$$

$$C\left(y^2 + \frac{Bx + E}{C}\,y\right) = -Ax^2 - Dx - F$$

$$C\left[y^2 + \frac{Bx + E}{C}\,y + \frac{(Bx + E)^2}{4C^2}\right] = -Ax^2 - Dx - F + \frac{(Bx + E)^2}{4C}$$

$$\left(y + \frac{Bx + E}{2C}\right)^2 = \frac{(Bx + E)^2}{4C^2} - \frac{Ax^2 + Dx + F}{C}$$

$$y + \frac{Bx + E}{2C} = \pm \sqrt{\frac{(Bx + E)^2}{4C^2} - \frac{Ax^2 + Dx + F}{C}}$$

$$y = \pm \sqrt{\frac{(Bx + E)^2}{4C^2} - \frac{Ax^2 + Dx + F}{C}} - \frac{Bx + E}{2C}$$

In BASIC this becomes

```
Y1 = SQR((B*X+E)^2/(4*C^2)-(A*X^2+D*X+F)/C)-((B*X+E)/(2*C)))
Y2 =-SQR((B*X+E)^2/(4*C^2)-(A*X^2+D*X+F)/C)-((B*X+E)/(2*C)))
```

The following is a complete listing of PROG. 5.1 which will graph conic sections with inputs A,B,C,D,E, and F.

PROG. 5.1

```
100  REM  GENERAL EQUATION PROGRAM
110  REM
120  TEXT
130  HOME
140  REM
150  REM  ACCEPT VALUES FOR THE COEFFICIENTS
160  REM
170  HOME
180  INPUT "DO YOU WISH TO USE THE DEFAULT VALUES   FOR THE
COEFFICIENTS? ";ANS$
190  IF  LEFT$ (ANS$,1) = "N" THEN 220
200 A = 1:B = 0:C = 1:D = 100:E =  - 80:F = 4100
210  GOTO 340
220  PRINT
230  PRINT "    AX^2+BXY+CY^2+DX+EY+F=0"
240  PRINT
250  INPUT "A = ";A
260  INPUT "B = ";B
270  INPUT "C = ";C
280  INPUT "D = ";D
290  INPUT "E = ";E
300  INPUT "F = ";F
310  REM
320  REM  DEFINE FUNCTION
```

PROG. 5.1 CONT.

```
330  REM
340  DEF  FN D(X) = (B * X + E) ^ 2 / (4 * C ^ 2) - (A * X ^
 2 + D * X + F) / C
350  REM
360  REM  GRAPHING ROUTINE
370  REM
380  REM
390  REM  ASSIGN SCREEN LIMITS
400  REM
410 HL = 279
420 VL = 159
430 VS = 80
440 HS = 140
450  REM
460  REM ACCEPT VALUES FOR XMIN, XMAX, YMIN, YMAX
470  REM
480  INPUT "USE DEFAULT VALUES FOR X AND Y RANGES?";ANS$
490  IF  LEFT$ (ANS$,1) = "N" THEN 520
500 X1 =  - 100:X2 = 100:Y1 =  - 100:Y2 = 100
510  GOTO 560
520  INPUT "XMIN = ";X1
530  INPUT "XMAX = ";X2
540  INPUT "YMIN = ";Y1
550  INPUT "YMAX = ";Y2
560  REM
570  REM  ASSIGN MULTIPLIERS AND DELTA X
580  REM
590 XM = 1
600 YM =  - .85
610 DX = 1
620  INPUT "HARD COPY?";Q$
630  IF Q$ <  > "" THEN  PR# 3
640  PRINT "     "A;"X^2 + "B;"XY + ";C;"Y^2 + ";D;"X + "E;"
Y "F" =0"
650  REM
660  REM  DRAW COORDINATE AXIS
670  REM
680  HGR
690  HCOLOR= 6
700  HPLOT 140,0 TO 140,159
710  HPLOT 0,80 TO 279,80
```

ROG. 5.1 CONT.

```
720  REM
730  REM  PLOTTING LOOP
740  REM
750  HCOLOR= 3
760  FOR X = X1 TO X2 STEP DX
770 XP = XM * X + HS
780  IF  FN D(X) < 0 THEN 860
790 YP = YM *  SQR ( FN D(X)) + (B * X + E) / (2 * C) + VS
800 YN =  - YM *  SQR ( FN D(X)) + (B * X + E) / (2 * C) + V
S
810  IF XP > HL OR XP < 0 THEN 860
820  IF YP > VL OR YP < 0 THEN 840
830  HPLOT XP,YP
840  IF YN > VL OR YN < 0 THEN 860
850  HPLOT XP,YN
860  NEXT X
870  END
```

GRAPH 5.1.1 ---ELLIPSE---

EQUATION: $4X^2+0XY+1Y^2+0X+0Y-4000=0$



GRAPH 5.1.2 ---CIRCLE---

EQUATION: $1X^2+0XY+1Y^2+100X-80Y+3800=0$



GRAPH 5.1.3 ---POINT---

EQUATION: $1X^2+0XY+1Y^2+100X-80Y+4100=0$



GRAPH 5.1.4 ---EMPTY---

EQUATION: $1X^2+2XY+Y^2+0X+0Y+49=0$



GRAPH 5.2.1 ---HYPERBOLA---

EQUATION: $1X^2+0XY-1Y^2-20X-20Y-100=0$



GRAPH 5.2.2 ---PAIR OF INTERSECTING LINES---

EQUATION: $1X^2+0XY-1Y^2-20X-20Y=0$



GRAPH 5.3.1 ---PARABOLA---

EQUATION: $1X^2+2XY+1Y^2+100X-80Y+3000=0$



GRAPH 5.3.2 ---LINE---

EQUATION: $1X^2+2XY+1Y^2+0X+0Y+0=0$



GRAPH 5.3.3 ---PARALLEL LINES---

EQUATION: $1X^2+2XY+1Y^2+0X+0Y-100=0$



GRAPH 5.3.4 ---EMPTY---

EQUATION: $1X^2+0XY+1Y^2+100X-80Y+4500=0$



6. TRANSITIONS

The next topic to consider is a closer look at the 3 cases of the general equation that were introduced in the last chapter. This chapter will investigate what happen as the discrimant of EQ. 5.1 approaches 0. To accomplish this in the first program, let B = 0, C = 1, D = 20, E = 0,F=-500, and let A range from -1 to 1 in steps of .25. As A aproach 0 the conics change from a hyperbola, when A > 0, to a parabola when A = 0, to an ellipse when A < 0. Graphs 6.1.1 through 6.1.9 illustrate the transition. Graph 6.1.10 is an overlay of this transition.

EQ. 5.1---- $Ax^2+Bxy+Cy^2+Dx+Ey+F=0$

GRAPH 6.1.1 A = -1



EQ. 5.1---- $Ax^2+Bxy+Cy^2+Dx+Ey+F=0$

GRAPH 6.1.2 A = -.75



GRAPH 6.1.3 A = -.50



EQ. 5.1---- $Ax^2+Bxy+Cy^2+Dx+Ey+F=0$

GRAPH 6.1.4 A = -.25



GRAPH 6.1.5 A = 0



EQ. 5.1---- $Ax^2+Bxy+Cy^2+Dx+Ey+F=0$

GRAPH 6.1.6 A = .25



GRAPH 6.1.7 A = .50



EQ. 5.1---- $Ax^2+Bxy+Cy^2+Dx+Ey+F=0$

GRAPH 6.1.8 A = .75



GRAPH 6.1.9 A = 1



EQ. 5.1---$Ax^2+Bxy+Cy^2+Dx+Ey+F=0$

GRAPH 6.1.10 COMPOSITE GRAPH AS A RANGES FROM -1 TO 1 STEP .25



The next program will let A=1,B=0,D=0,E=20,F=-500. and let C range from -1 to 1. Graphs 6.2.1 to 6.2.5 show the transition of the conics. Graph 6.2.6 is an overlay of the previous graphs.

$$Ax^2+Bxy+Cy^2+Dx+Ey+F=0$$

GRAPH 6.2.1 C = -1



GRAPH 6.2.2 C = -.50



$Ax^2+Bxy+Cy^2+Dx+Ey+F=0$

GRAPH 6.2.3 C = 0



GRAPH 6.2.4 C = .50



$Ax^2+Bxy+Cy^2+Dx+Ey+F=0$

GRAPH 6.2.5 C = 1



GRAPH 6.2.6 COMPOSITE GRAPH AS C RANGES FROM -1 TO 1 STEP .25



The final progam in this chapter investigates the graphs of the general equation as B is allowed to range from -3.5 to 3.5 in steps of .50. The other coefficients are held constant with values of: A=1,C=1,D=20,E=20,and F=-500. Graphs 6.3.1 to 6.3.15 illustrate the transition from conic to conic. Graph 6.3.16 is a composite of the B range.

$$Ax^2+Bxy+Cy^2+Dx+Ey+F=0$$

GRAPH 6.3.1 B = -3.5



GRAPH 6.3.2 B = -3



$Ax^2+Bxy+Cy^2+Dx+Ey+F=0$

GRAPH 6.3.3 B = -2.5



GRAPH 6.3.4 B = -2



$Ax^2+Bxy+Cy^2+Dx+Ey+F=0$

GRAPH 6.3.5 B = -1.5



GRAPH 6.3.6 B = -1



$Ax^2+Bxy+Cy^2+Dx+Ey+F=0$

GRAPH 6.3.7 B = -.5



GRAPH 6.3.8 B = 0



$Ax^2+Bxy+Cy^2+Dx+Ey+F=0$

GRAPH 6.3.9 B = .5



GRAPH 6.3.10 B = 1



$Ax^2+Bxy+Cy^2+Dx+Ey+F=0$

GRAPH 6.3.11 B = 1.5



GRAPH 6.3.12 B = 2



$$Ax^2+Bxy+Cy^2+Dx+Ey+F=0$$

GRAPH 6.3.13 B = 2.5



GRAPH 6.3.14 B = 3



$Ax^2+Bxy+Cy^2+Dx+Ey+F=0$

GRAPH 6.3.15 B = 3.5



GRAPH 6.3.16 COMPOSITE GRAPH AS B RANGES FROM -3.5 TO 3.5 STEP .25



The following 3 programs are the complete listings of the programs that created the 3 sets of graphs.

PROG. 6.1

```
100  REM  BURST---A
110  REM  ASSIGN SCREEN LIMITS
120 HL = 279:VL = 159:VS = 80:HS = 140
130  REM  SET DEFAULT VALUE FOR XMIN, XMAX, VMIN, AND VMAX
140 X1 =  - 120:X2 = 135:Y1 =  - 100:Y2 = 100
150  REM  ASSIGN COEFFICIENTS
160 B = 0:C = 1:D = 20:E = 0:F =  - 500
170  REM  ASSIGN MULTIPLIERS AND DELTA X
180 XM = 1
190 YM =  - .85
200 DX = 1
210  REM  DRAW COORDINATE AXIS
220  HGR
230  HCOLOR= 5
240  HPLOT 141,0 TO 141,159
250  HPLOT 0,80 TO 279,80
260  REM  PLOTTING LOOP
270  FOR A =  - 1 TO 1 STEP .25
280  IF A < 0 THEN CL = 6
290  IF A = 0 THEN CL = 2
300  IF A > 0 THEN CL = 7
310  HCOLOR= CL
320  FOR X = X1 TO X2 STEP DX
330 XP = XM * X + HS
340  IF XP > HL OR XP < 0 THEN 420
350  IF (B * X + E) ^ 2 / (4 * C ^ 2) - (A * X ^ 2 + D * X +
 F) / C < 0 THEN 420
360 YP = YM * ( SQR ((B * X + E) ^ 2 / (4 * C ^ 2) - (A * X
^ 2 + D * X + F) / C) - ((B * X + E) / (2 * C))) + VS
370  IF YP > VL OR YP < 0 THEN 390
380  HPLOT XP,YP
390 YN = YM * ( -  SQR ((B * X + E) ^ 2 / (4 * C ^ 2) - (A *
 X ^ 2 + D * X + F) / C) - ((B * X + E) / (2 * C))) + VS
400  IF YN > VL OR YN < 0 THEN 420
410  HPLOT XP,YN
420  NEXT X
430  NEXT A
440  END
```

PROG. 6.2

```
100  REM  BURST---B
110  REM  ASSIGN SCREEN LIMITS
120 HL = 279:VL = 159:VS = 80:HS = 140
130  REM  SET DEFAULT VALUE FOR XMIN, XMAX, VMIN, AND VMAX
140 X1 =  - 120:X2 = 135:Y1 =  - 100:Y2 = 100
150  REM  ASSIGN COEFFICIENTS
160 A = 1:C = 1:D = 20:E = 20:F =  - 500
170  REM  ASSIGN MULTIPLIERS AND DELTA X
180 XM = 1
190 YM =  - .85
200 DX = 1
210  REM  DRAW COORDINATE AXIS
220  HGR
230  HCOLOR= 5
240  HPLOT 141,0 TO 141,159
250  HPLOT 0,80 TO 279,80
260  REM  PLOTTING LOOP
270  FOR B =  - 3.5 TO 3.5 STEP .25
280  FOR X = X1 TO X2 STEP DX
290 XP = XM * X + HS
300  IF XP > HL OR XP < 0 THEN 380
310  IF (B * X + E) ^ 2 / (4 * C ^ 2) - (A * X ^ 2 + D * X +
 F) / C < 0 THEN 380
320 YP = YM * ( SQR ((B * X + E) ^ 2 / (4 * C ^ 2) - (A * X
^ 2 + D * X + F) / C) - ((B * X + E) / (2 * C))) + VS
330  IF YP > VL OR YP < 0 THEN 350
340  HPLOT XP,YP
350 YN = YM * ( -  SQR ((B * X + E) ^ 2 / (4 * C ^ 2) - (A *
 X ^ 2 + D * X + F) / C) - ((B * X + E) / (2 * C))) + VS
360  IF YN > VL OR YN < 0 THEN 380
370  HPLOT XP,YN
380  NEXT X
390  NEXT B
400  END
```

PROG. 6.3

```
100  REM  BURST---C
110  REM  ASSIGN SCREEN LIMITS
120 HL = 279:VL = 159:VS = 80:HS = 140
130  REM  SET DEFAULT VALUE FOR XMIN, XMAX, VMIN, AND VMAX
140 X1 =  - 120:X2 = 135:Y1 =  - 100:Y2 = 100
150  REM  ASSIGN COEFFICIENTS
160 A = 1:B = 0:D = 0:E = 20:F =  - 500
170  REM  ASSIGN MULTIPLIERS AND DELTA X
180 XM = 1
190 YM =  - .85
200 DX = 1
210  REM  DRAW COORDINATE AXIS
220  HGR
230  HCOLOR= 5
240  HPLOT 141,0 TO 141,159
250  HPLOT 0,80 TO 279,80
260  REM  PLOTTING LOOP
270  FOR C =  - 1 TO 1 STEP .25
280  FOR X = X1 TO X2 STEP DX
290  IF C = 0 THEN C = .001
300 XP = XM * X + HS
310  IF XP > HL OR XP < 0 THEN 390
320  IF (B * X + E) ^ 2 / (4 * C ^ 2) - (A * X ^ 2 + D * X +
 F) / C < 0 THEN 390
330 YP = YM * ( SQR ((B * X + E) ^ 2 / (4 * C ^ 2) - (A * X
^ 2 + D * X + F) / C) - ((B * X + E) / (2 * C))) + VS
340  IF YP > VL OR YP < 0 THEN 360
350  HPLOT XP,YP
360 YN = YM * ( -  SQR ((B * X + E) ^ 2 / (4 * C ^ 2) - (A *
 X ^ 2 + D * X + F) / C) - ((B * X + E) / (2 * C))) + VS
370  IF YN > VL OR YN < 0 THEN 390
380  HPLOT XP,YN
390  NEXT X
400  NEXT C
410  END
```

7. ECCENTRICITY

Another point of view to investigate conic sections is to define the conic in terms of its eccentricity. Consider the following definition of a conic section:

DEFINITION 7.1 Let $e>0$ be a fixed number, let D be a fixed line, and O be a fixed point not on the line D. Let $|PR|$ denote the distance of the point $P(x,y)$ from the line D. Finally let G be the collection of all points P such that

$$\frac{|OP|}{|RP|} = e$$

Then it can be shown G is a conic section. If $e=1$ then G is a parabola. If $0<e<1$, then G is an ellipse. If $e>1$ then then G is a hyperbola [1. p.330].

A useful application now is to use this definition to write an equation for the conic in polar coordinates. In order to obtain a simple result we put the focus F at the origin and we let the directrix be a vertical line d units ($d>0$) to the left of the focus, as shown in FIG. 7.1. Let $P(r,\theta)$ be any point on the conic to the right of the directrix and on the terminal side of θ, and draw perpendiculars PQ and PR to the principal axis and the directrix, restectively. By DEFINITION 7.1, the point P is on the conic if and only if $|OP| = e|RP|$ EQ. 7.1.

FIG. 7.1



Since P is to the right of the directrix, RP>0 and so |RP|=RP. |OP|=r because r>0 since P is on the terminal side of θ . So from EQ. 7.1 we have r=e(RP) EQ. 7.2. However RP=DQ=DO+OQ=d+rcosθ . Subatituting this expression for RP in EQ 7.2 we obtain

$$\text{EQ.7.3} \qquad r = \frac{ed}{1-e\cos\theta}$$

In a similar manner we may derive an equation of a conic if the directrix corresponding to the focus at the pole is to the right of the focus and we obtain

$$\text{EQ. 7.4} \qquad r = \frac{ed}{1+e\cos\theta}$$

[2. pp. 562-563]

Using the equations 2.3.1 and 2.3.2 with EQ. 7.3 we now can convert the polar equations to cartesian coordinates and then to BASIC. This becomes:

```
X=(E*D/(1+E*COS(TH)))*COS(TH)
Y=(E*D/(1+E*COS(TH)))*SIN(TH)
```

Graphs 7.1 to 7.10 illustrate the affect the eccentricity has EQ. 7.3.

GRAPH 7.1 D=10, E=.25



GRAPH 7.2 D=10, E=.50



GRAPH 7.3 D=10, E=.75



GRAPH 7.4 D=10, E=1



GRAPH 7.5 D=10, E=1.25



GRAPH 7.6 D=10, E=1.50



GRAPH 7.7 D=10, E=2



GRAPH 7.8 D=10, E=4



GRAPH 7.9 D=10, E=6



GRAPH 7.10 D=10, E=8



The following is a complete listing of PROG. 7.1 that generated the eccentricity graphs.

PROG. 7.1

```
100  REM  ECCENTRICITY PROGRAM
110  REM
120  TEXT
130  HOME
140  REM
150  REM  ACCEPT INPUTS FOR THE PARAMETERS
160  REM
170  HOME
180  PRINT
190  PRINT "           2ED"
200  PRINT "   R = ----------"
210  PRINT "       1-ECOS(TH)"
220  PRINT
230  INPUT "ENTER VALUE FOR D  ";D
240  REM
```

PROG. 7.1 CONT.

```
250  REM  DEFINE FUNCTION
260  REM
270  DEF  FN X(TH) = (2 * E * D / (1 - E *  COS (TH))) *  CO
S (TH)
280  DEF  FN Y(TH) = (2 * E * D / (1 - E *  COS (TH))) *  SI
N (TH)
290  REM
300  REM  GRAPHING ROUTINE
310  REM
320  REM
330  REM  ASSIGN SCREEN LIMITS
340  REM
350 HL = 279
360 VL = 159
370 VS = 80
380 HS = 140
390  REM
400  REM  ACCEPT VALUES FOR THMIN,THMAX
410  REM
420  INPUT "USE DEFAULT VALUES FOR X AND Y RANGES?";ANS$
430  IF  LEFT$ (ANS$,1) = "N" THEN 470
440 T1 = 0:T2 = 6.28
450  REM
460  REM  ASSIGN MULTIPLIERS AND DELTA THEATA
470  REM
480 XM = 1
490 YM =  - .85
500 DT = .05
510  FOR E = .25 TO 1 STEP .25
520  PR# 3
530  PRINT "D = ";D;"   E = ";E
540  REM
550  REM  DRAW COORDINATE AXIS
560  REM
570  HGR
580  HCOLOR= 6
590  HPLOT 140,0 TO 140,159
600  HPLOT 0,80 TO 279,80
610  REM
620  REM  PLOTTING LOOP
630  REM
640  HCOLOR= 3
```

PROG. 7.1 CONT.

```
650  FOR TH = T1 TO T2 STEP DT
660  IF TH = 0 THEN 720
670 XP = XM *  FN X(TH) + HS
680 YP = YM *  FN Y(TH) + VS
690  IF XP > HL OR XP < 0 THEN 720
700  IF YP > VL OR YP < 0 THEN 720
710  HPLOT XP,YP
720  NEXT TH
730  NEXT E
740  END
```

8. SUMMARY

This thesis was an attempt to provide some useful ideas in graphing conic sections. It is hoped that through the use of some of the ideas presented, the microcomputer can be as an instructional aid to better understand conics. A comment about the transition programs is appropriate here. It should be noted that smaller increments around pivot points (i.e. A=0 in PROG. 6.1) would provide some interesting graphs as the transition occurs. However, one would have to change the plotting limits to accommodate the range of the curves. I would also like to list some other sources that were useful to me when I was getting my ideas for this thesis.

The textbook that I use in high school is MODERN ALGEBRA AND TRIGONOMETRY BOOK 2, by Dolciani, Berman, and Wooton. The publisher is Houghton Mifflin. Two good books on microcomputer graphics that I was able to obtain were: MICROCOMPUTER GRAPHICS, by Roy E. Myers, publisher Addison Wesley, and GRAPHIC SOFTWARE FOR MICROCOMPUTERS, by B. J. Korites, publisher Kern Publications. These books along with the texts listed in the bibliography were very valuable sources in writing this thesis.